```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

We consider the following matrix of differential operators taken from A. D. Polyanin, A. V. Manzhirov, *Handbook of Mathematics for Engineers and Scientists*, Chapman, 2007:

```
> R[1]:=evalm([[d[t]-k*d[x]-a[1],-b[1]],[-a[2],d[t]-k*d[x]-b[2]]]);
```

$$R_1 := \begin{bmatrix} d_t - k\,d_x - a_1 & -b_1 \\ -a_2 & d_t - k\,d_x - b_2 \end{bmatrix}$$

We denote by $A = \mathbb{Q}(a_1, a_2, b_1, b_2, k)[d_t, d_x]$ the commutative polynomial ring of differential operators in d_t and d_x with coefficients in the field $\mathbb{Q}(a_1, a_2, b_1, b_2, k)$

```
> A:=DefineOreAlgebra(diff=[d[t],t],diff=[d[x],x],polynom=[t,x],comm=[a[1],
> a[2],b[1],b[2],k]):
```

and $M_1 = A^{1\times 2}/(A^{1\times 2}\,R_1)$ the A-module finitely presented by the matrix R_1. The endomorphism ring $E_1 = \mathrm{end}_A(M_1)$ is defined by the generators f_i's defined by $f_i(\pi_1(\lambda)) = \pi_1(\lambda\,P_i)$, where $\pi_1 : A^{1\times 2} \longrightarrow M_1$ denote the projection onto M_1, λ is any element of $A^{1\times 2}$ and P_i is one of the following two matrices:

```
> Endo[1]:=MorphismsConstCoeff(R[1],R[1],A):
```

```
> Endo[1][1];
```

$$[\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & b_1 \\ a_2 & -a_1 + b_2 \end{bmatrix}]$$

The family of generators $\{f_1 = \mathrm{id}_{M_1}, f_2\}$ of the A-module E_1 satisfy the A-linear relations defined by $Endo[1][2]\,F = 0$, where $F = (f_1 \quad f_2)^T$ and $Endo[1][2]$ is the following matrix:

```
> Endo[1][2];
```

$$\begin{bmatrix} -a_2\,b_1 & d_t - k\,d_x - b_2 \\ d_t - k\,d_x - a_1 & -1 \end{bmatrix}$$

Let us study the existence of idempotents of the ring E_1 defined by constant matrices:

```
> Idem[1]:=IdempotentsMatConstCoeff(R[1],Endo[1][1],A,0,alpha):
> Idem[1][1];
```

$$[\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 1/2\,\alpha_1\,a_1 - 1/2\,\alpha_1\,b_2 + 1/2 & \alpha_1\,b_1 \\ \alpha_1\,a_2 & -1/2\,\alpha_1\,a_1 + 1/2\,\alpha_1\,b_2 + 1/2 \end{bmatrix}]$$

We obtain three idempotents of E_1 defined by means of constant matrices. In particular, the first two ones are the trivial ones 0 and id_M. The last one is defined by a matrix $Idem[1][1,3]$ whose entries belong to the ring $B = Q(a_1, a_2, b_1, b_2, k)[\alpha_1]/(((a_1 - b_2)^2 + 4\,a_2\,b_1)\,\alpha_1^2 - 1)[d_t, d_x]$:

```
> B:=Idem[1][2]: collect(B[9][1],alpha[1]);
```

$$-1 + \left(a_1{}^2 - 2\,a_1\,b_2 + b_2{}^2 + 4\,a_2\,b_1\right)\alpha_1{}^2$$

Using the fact that B is a commutative polynomial ring over a field, we know that the matrix R_1 is then equivalent to the block diagonal matrix $T_1 = V_1\,R_1\,U_1^{-1}$ defined by:

```
> S[1]:=simplify(subs(alpha[1]^2=1/((a[1]-b[2])^2+4*a[2]*b[1]),
> HeuristicDecomposition(R[1],Idem[1][1,3],B))):
```

```
> T[1]:=map(collect,S[1][1],{d[t],d[x]},distributed);
```

$$T_1 := \begin{bmatrix} d_t - k\,d_x - \frac{\alpha_1\,a_1+\alpha_1\,b_2-1}{2\,\alpha_1} & 0 \\ 0 & d_t - k\,d_x - \frac{\alpha_1\,b_2+\alpha_1\,a_1+1}{2\,\alpha_1} \end{bmatrix}$$

The unimodular matrices U_1 and V_1 are then defined by:

```
> U[1]:=evalm(S[1][2]); V[1]:=evalm(S[1][3]);
```

$$U_1 := \begin{bmatrix} -2\,\alpha_1\,a_2 & -\alpha_1\,b_2+\alpha_1\,a_1+1 \\ 2\,\alpha_1\,a_2 & 1-\alpha_1\,a_1+\alpha_1\,b_2 \end{bmatrix} \quad V_1 := \begin{bmatrix} -2\,\alpha_1\,a_2 & -\alpha_1\,b_2+\alpha_1\,a_1+1 \\ 2\,\alpha_1\,a_2 & 1-\alpha_1\,a_1+\alpha_1\,b_2 \end{bmatrix}$$

We now consider a second example coming from A. D. Polyanin, A. V. Manzhirov, *Handbook of Mathematics for Engineers and Scientists*, Chapman, 2007, which is defined by the following matrix of differential operators:

```
> R[2]:=evalm([[d[t]-k*d[x]^2-a[1],-b[1]],[-a[2],d[t]-k*d[x]^2-b[2]]]);
```

$$R_2 := \begin{bmatrix} d_t - k\,{d_x}^2 - a_1 & -b_1 \\ -a_2 & d_t - k\,{d_x}^2 - b_2 \end{bmatrix}$$

Doing similarly as before, we obtain that the endomorphism ring $E_2 = \mathrm{end}_A(M_2)$ of the A-module $M_2 = A^{1\times 2}/(A^{1\times 2}\,R_2)$ is generated by two generators g_i's defined by $g_i(\pi_2(\lambda)) = \pi_2(\lambda\,P_i)$, where $\pi_2 : A^{1\times 2} \longrightarrow M_2$ denote the projection onto M_2, λ is any element of $A^{1\times 2}$ and P_i is one of the following two matrices:

```
> Endo[2]:=MorphismsConstCoeff(R[2],R[2],A):
```

```
> Endo[2][1];
```

$$[\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & b_1 \\ a_2 & -a_1+b_2 \end{bmatrix}]$$

The family of generators $\{g_1 = \mathrm{id}_{M_2}, g_2\}$ of the A-module E_2 satisfy the A-linear relations defined by $Endo[2][2]\,G = 0$, where $G = (g_1 \quad g_2)^T$ and $Endo[2][2]$ is the following matrix:

```
> Endo[2][2];
```

$$\begin{bmatrix} a_2\,b_1 & -d_t + k\,{d_x}^2 + b_2 \\ -d_t + k\,{d_x}^2 + a_1 & 1 \end{bmatrix}$$

We can now study the existence of idempotents of the ring E_2 defined by constant matrices:

```
> Idem[2]:=IdempotentsMatConstCoeff(R[2],Endo[2][1],A,0,alpha):
```

```
> Idem[2][1];
```

$$[\begin{bmatrix} 1/2\,\alpha_1\,a_1 - 1/2\,\alpha_1\,b_2 + 1/2 & \alpha_1\,b_1 \\ \alpha_1\,a_2 & -1/2\,\alpha_1\,a_1 + 1/2\,\alpha_1\,b_2 + 1/2 \end{bmatrix}, \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}]$$

We obtain three idempotents of E_2 defined by means of constant matrices: 0, id_M and an idempotent e defined by the matrix $Idem[2][1,1] \in B^{2\times 2}$, i.e., $e \in \mathrm{end}_B(B \otimes_A M_2)$:

```
> B:=Idem[2][2]: collect(B[9][1],alpha[1]);
```

$$-1 + \left({a_1}^2 - 2\,a_1\,b_2 + {b_2}^2 + 4\,a_2\,b_1\right){\alpha_1}^2$$

Then, the matrix R_2 is equivalent to the block-diagonal matrix $T_2 = V_2\,R_2\,U_2^{-1}$ defined by:

```
> S[2]:=simplify(subs(alpha[1]^2=1/((a[1]-b[2])^2+4*a[2]*b[1]),
> HeuristicDecomposition(R[2],Idem[2][1,1],B))):
```

```
> T[2]:=map(collect,S[2][1],{d[t],d[x]},distributed);
```

$$T_2 := \begin{bmatrix} d_t - k\,{d_x}^2 - \frac{\alpha_1\,a_1+\alpha_1\,b_2-1}{2\,\alpha_1} & 0 \\ 0 & d_t - k\,{d_x}^2 - \frac{\alpha_1\,b_2+\alpha_1\,a_1+1}{2\,\alpha_1} \end{bmatrix}$$

The unimodular matrices U_2 and V_2 are then defined by:

```
> U[2]:=evalm(S[2][2]); V[2]:=evalm(S[2][3]);
```

$$U_2 := \begin{bmatrix} 2\,\alpha_1\,a_2 & \alpha_1\,b_2 - \alpha_1\,a_1 - 1 \\ -2\,\alpha_1 a_2 & -1 + \alpha_1\,a_1 - \alpha_1\,b_2 \end{bmatrix} \quad V_2 := \begin{bmatrix} 2\,\alpha_1\,a_2 & \alpha_1\,b_2 - \alpha_1\,a_1 - 1 \\ -2\,\alpha_1\,a_2 & -1 + \alpha_1\,a_1 - \alpha_1\,b_2 \end{bmatrix}$$

To finish, we consider the matrix of differential operators defined by:

```
> R[3]:=evalm([[d[t]^2-k*d[x]^2-a[1],-b[1]],[-a[2],d[t]^2-k*d[x]^2-b[2]]]);
```

$$R_3 := \begin{bmatrix} {d_t}^2 - k\,{d_x}^2 - a_1 & -b_1 \\ -a_2 & {d_t}^2 - k\,{d_x}^2 - b_2 \end{bmatrix}$$

The endomorphism ring $E_3 = \mathrm{end}_A(M_3)$ of the A-module $M_3 = A^{1\times 2}/(A^{1\times 2}\,R_3)$ is generated by two generators h_1 and h_2 respectively defined by means of the following matrices:

```
> Endo[3]:=MorphismsConstCoeff(R[3],R[3],A):
```

```
> Endo[3][1];
```

$$[\begin{bmatrix} 0 & b_1 \\ a_2 & -a_1 + b_2 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}]$$

The family of generators $\{g_1, g_2 = \mathrm{id}_{M_3}\}$ of the A-module E_3 satisfy the A-linear relations $Endo[3][2]\,G = 0$, where $G = (g_1 \quad g_2)^T$ and the matrix $Endo[3][2]$ is defined by:

```
> Endo[3][2];
```

$$\begin{bmatrix} -1 & {d_t}^2 - k\,{d_x}^2 - a_1 \\ {d_t}^2 - k\,{d_x}^2 - b_2 & -a_2 b_1 \end{bmatrix}$$

Let us compute idempotents of the ring E_3 which are defined by means of constant matrices:

```
> Idem[3]:=IdempotentsMatConstCoeff(R[3],Endo[3][1],A,0,alpha):
```

```
> Idem[3][1];
```

$$[\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 1/2\,\alpha_1\,a_1 - 1/2\,\alpha_1\,b_2 + 1/2 & \alpha_1\,b_1 \\ \alpha_1\,a_2 & -1/2\,\alpha_1\,a_1 + 1/2\,\alpha_1\,b_2 + 1/2 \end{bmatrix}]$$

We obtain the two trivial idempotents 0 and id_{M_3} of E_3 but a non-trivial one defined by the third matrix $Idem[3][1,3]$ of $Idem[3]$. The entries of $Idem[3][1,3]$ also belong to the ring B:

```
> B:=Idem[3][2]: collect(B[9][1],alpha[1]);
```

$$-1 + \left({a_1}^2 - 2\,a_1\,b_2 + {b_2}^2 + 4\,a_2\,b_1\right){\alpha_1}^2$$

Then, the matrix R_3 is equivalent to the block-diagonal matrix $T_3 = V_3\,R_3\,U_3^{-1}$ defined by:

```
> S[3]:=simplify(subs(alpha[1]^2=1/((a[1]-b[2])^2+4*a[2]*b[1]),
> HeuristicDecomposition(R[3],Idem[3][1,3],B))):
```

```
> T[3]:=map(collect,S[3][1],{d[t],d[x]},distributed);
```

$$T_3 := \begin{bmatrix} {d_t}^2 - k\,{d_x}^2 - \frac{\alpha_1\,a_1+\alpha_1\,b_2-1}{2\,\alpha_1} & 0 \\ 0 & {d_t}^2 - k\,{d_x}^2 - \frac{\alpha_1\,b_2+\alpha_1\,a_1+1}{2\,\alpha_1} \end{bmatrix}$$

The unimodular matrices U_3 and V_3 are then defined by:

```
> U[3]:=evalm(S[3][2]); V[3]:=evalm(S[3][3]);
```

$$U_3 := \begin{bmatrix} 2\,\alpha_1\,a_2 & \alpha_1\,b_2 - \alpha_1\,a_1 - 1 \\ 2\,\alpha_1\,a_2 & 1 - \alpha_1\,a_1 + \alpha_1\,b_2 \end{bmatrix} \quad V_3 := \begin{bmatrix} 2\,\alpha_1\,a_2 & \alpha_1\,b_2 - \alpha_1\,a_1 - 1 \\ 2\,\alpha_1\,a_2 & 1 - \alpha_1\,a_1 + \alpha_1\,b_2 \end{bmatrix}$$